QC
874.3
.U63
no.2

NOAA Eastern Region Computer Programs
and Problems NWS ERCP - No. 2

AN AFOS APPLICATIONS PROGRAM TO COMPUTE THREE-HOURLY STREAM STAGES

Scientific Services Division
Eastern Region Headquarters
September 1981

CENTRAL
LIBRARY
PER
DEC 1981
N.O.A.A.
U. S. Dept. of Commerce

U.S. DEPARTMENT OF COMMERCE / National Oceanic and Atmospheric Administration / National Weather Service

NOAA TECHNICAL MEMORANDUM
National Weather Service, Eastern Region Computer Programs and Problems

The Eastern Region Computer Programs and Problems (ERCP) series is a subset of the Eastern Region Technical Memorandum series. It will serve as the vehicle for the transfer of information about fully documented AFOS application programs. The format of ERCP - No. 1 will serve as the model for future issuances in this series.

NOAA Eastern Region Computer Programs and Problems NWS ERCP

1 An AFOS Version of the Flash Flood Checklist.

81 3571

QC
874.3
U63
no.2

NOAA Eastern Region Computer Programs and Problems - No. 2

Alan P. Blackburn
WSFO Buffalo, New York

Scientific Services Division
Eastern Region Headquarters
September 1981

An AFOS applications program to compute three-hourly stream stages.

UNITED STATES
DEPARTMENT OF COMMERCE
Malcolm Baldrige, Secretary

NATIONAL OCEANIC AND
ATMOSPHERIC ADMINISTRATION
James P. Walsh, Acting Administrator

National Weather
Service
Richard E. Hallgren, Director

81 3571

CENTRAL
LIBRARY
DEC 8 1981
N.O.A.A.
U. S. Dept. of Commerce

AN AFOS APPLICATIONS PROGRAM TO COMPUTE THREE-HOURLY STREAM STAGES

Alan P. Blackburn
WSFO Buffalo, New York

I. General Information

A. Summary:

This ERCP describes a hydrology program, HYDRO, developed at WSFO Buffalo. It computes 3-hour stages for selected stream gages in western New York State, but can be adapted to any location. HYDRO enables forecasters to easily compute stream stages if heavy precipitation occurs and a hydrologist is not available.

B. Environment:

This Basic program runs on the Eclipse S/230 minicomputer via the Data General Basic interpreter (Rev. 4.10, 8/31/76).

C. References:

Basic Basic -- Data General manual 093-000088
Extended Basic User's Manual -- Data General Manual 093-000141-00

II. Application

A. Complete Program Description:

The purpose of the program is to compute 3-hour stage forecasts for selected stream points. The program uses four 6-hour precipitation amounts (24 hours' worth), the river stage at precipitation onset (base stage) and a runoff index (flood index or antecedent index) for each point as input. Six-hour precipitation values were used because they correspond to LFM values, and it was felt that 3-hour totals would be difficult to estimate. The runoff index is obtained from the RFC's guidance and/or headwaters statements. A basin runoff index may be used if it is the same for all points in the basin.

The program divides the 6-hour precipitation values in half to obtain 3-hour amounts, then enters a precip-index table to compute the 3-hour runoff. (These tables, one for each basin used, are contained in the body of the program.) The runoff computation uses linear interpolation between points in the tables. Below is an example of how runoff is calculated for the period between 6 and 9 hours:

1. Find the total precipitation to 9 hours.

2. Enter a precip-index table and find the corresponding runoff.

3. Subtract the runoff for the 0-3 hour and 3-6 hour periods from the 9-hour runoff to obtain runoff for the 6-9 hour period.

The program next computes the flow due to the runoff by multiplying runoff by the point's three-hour hydrograph (also contained in the program body). (The three-hour unit hydrographs were derived from six-hour ones via a trial and error procedure. Precipitation amounts that yielded 1 inch of runoff were entered, then the 3-hour hydrograph was adjusted to give the same flow as the 6-hour hydrograph.) Flows are lagged and summed to obtain the total flow.

The program then uses the total flow to compute the stream stage for most of the points via rating tables (again, these are in the program body). Stages are linearly interpolated, high stages are linearly extrapolated from the last 2 interpolation points. This method is generally valid for a headwaters area.

For the lower Chemung basin, however, a different method of stage computation was used. A crest-stage relationship was developed between the stage at Erwins and that at points further downstream and then used to forecast those stages. (This method assumes uniform precipitation over the basin.)

B. Machine Requirements:

The Basic interpreter takes up around 14K of memory, so most WSO's could not run HYDRO with AFOS up. Disk storage runs around 52 blocks (26,486 bytes) for the HYDRO file itself. The run time will vary with the number of gages used, but once the data are entered 4-5 gages can be processed in about 3 minutes.

C. Structure of Software:

See the accompanying flow chart (Figure 1).

D. Data Base:

All I/O is via the Dasher and all other data referenced by HYDRO is contained in the program code. No other files are used, and there is no data base access.

III. Procedures

A. Initiation of Program:

The program must be placed in the Basic directory (BASIC.DR) with the MOVE command. After that, the program is run by first invoking the Basic interpreter (command: BASIC) and then typing the command to run the program (command: RUN "HYDRO").

B. Input Required:

The program requests entry of the precipitation amounts, base stages, runoff indexes and a choice of basins and gages on the Dasher. It also asks if all stages are desired, or only the high ones. See sample runs, Figures 2-5.

Figure 1


1. INPUT HIGH STAGE
2. INPUT PRECIP AMOUNTS
3. HIGH STAGE OR ALL STAGE? Q2$
4. INPUT FLOOD STAGES
SAME GAGES AS BEFORE?
YES
NO
SELECT BASIN I2
SELECT GAGE I3
INPUT FLOOD INDEX AND BASE STAGE
ANOTHER GAGE?
YES
NO
ANOTHER BASIN?
YES
NO
1. CONVERT PRECIP TO 3 HOURLY
2. COMPUTE RUNOFF FOR 3 HOUR PERIODS
3. COMPUTE BASE FLOW
4. COMPUTE 3 HOURLY FLOWS
5. COMPUTE STAGES
IS POINT IN LOWER CHEMUNG?
YES
NO
COMPUTE DOWNSTREAM STAGE
PRINT STAGES OR CREST
ANOTHER GAGE?
YES
NO
ANOTHER BASIN?
YES
NO
NEW PRECIP?
YES
NO
STOP

```
RUN\ ?
STOP AT 0360
*RUN

TYPE 4 PRECIP AMOUNTS (6 HRLY) BEGINNING WITH THE EARLIEST ? .5,1,.75,.4
IF YOU WANT ONLY HIGH STAGES AND CRESTS TYPE   HIGH.
 — IF YOU WANT ALL STAGES TYPE ALL ? ALL
WHICH BASIN DO YOU WANT FORECASTS FOR?
 1 = ALL
 2 = CHEMUNG  4 = BUFFALO AREA CREEKS
 3 = GENESEE  5 = ALLEGHENY
 ? 4
WHICH GAGE DO YOU WANT FORECASTS FOR?
   1 = ALL              23 = LANCASTER
  20 = BATAVIA          24 = GARDENVILLE
  21 = WILLIAMSVILLE    25 = EBENEZER
  22 = SCAJAQUADA       26 = GOWANDA
 ? 20
TYPE STAGE AT START OF PRECIP ? 3.55
TYPE RUNOFF INDEX AT LANCASTER ? 20
TONAWANDA CREEK AT BATAVIA   FLOOD STAGE 9 FEET

  BASE STAGE  3.55   FLOOD INDEX 18.18

  6 HRLY PRECIP  0.50  1.00  0.75  0.40

  TIMES OF STAGES ARE FROM THE START OF THE FIRST 6 HOUR PERIOD


  3 HOURS  3.55   27 HOURS  7.13   51 HOURS  8.16   75 HOURS  4.78   99 HOURS  3.86
  6 HOURS  3.55   30 HOURS  8.19   54 HOURS  7.48   78 HOURS  4.68  102 HOURS  3.73
  9 HOURS  3.57   33 HOURS  9.13*  57 HOURS  6.87   81 HOURS  4.54  105 HOURS  3.67
 12 HOURS  3.62   36 HOURS  9.74*  60 HOURS  6.33   84 HOURS  4.45  108 HOURS  3.60
 15 HOURS  3.82   39 HOURS  9.93*  63 HOURS  5.89   87 HOURS  4.34  111 HOURS  3.58
 18 HOURS  4.25   42 HOURS  9.78*  66 HOURS  5.52   90 HOURS  4.23  114 HOURS  3.55
 21 HOURS  5.01   45 HOURS  9.38*  69 HOURS  5.19   93 HOURS  4.11  117 HOURS  3.55
 24 HOURS  6.05   48 HOURS  8.82   72 HOURS  4.95   96 HOURS  3.97  120 HOURS  3.55


ANOTHER GAGE ?   Y OR N ?
```

Figure 2

Figure 3

```
RUN

 TYPE 4 PRECIP AMOUNTS (6 HRLY)  BEGINNING WITH THE EARLIEST ? .5,1,.75,.4
IF YOU WANT ONLY HIGH STAGES AND CRESTS TYPE  HIGH
   IF YOU WANT ALL STAGES TYPE ALL ? HIGH
WHICH BASIN DO YOU WANT FORECASTS FOR?
  1 == ALL
  2 = CHEMUNG  4 = BUFFALO AREA CREEKS
  3 = GENESEE  5 = ALLEGHENY
 ? 4
WHICH GAGE DO YOU WANT FORECASTS FOR?
   1 = ALL              23 = LANCASTER
  20 = BATAVIA          24 = GARDENVILLE
  21 = WILLIAMSVILLE    25 = EBENEZER
  22 = SCAJAQUADA       26 = GOWANDA
 ? 20
TYPE STAGE AT START OF PRECIP ? 3.55
TYPE RUNOFF INDEX AT LANCASTER ? 20
TONAWANDA CREEK AT BATAVIA   FLOOD STAGE 9 FEET

  BASE STAGE  3.55   FLOOD INDEX 18.18

  6 HRLY PRECIP  0.50  1.00  0.75  0.40

  TIMES OF STAGES ARE FROM THE START OF THE FIRST 6 HOUR PERIOD


  30 HOURS  8.19
  33 HOURS  9.13*
  36 HOURS  9.74*
  39 HOURS  9.93*
  42 HOURS  9.78*
  45 HOURS  9.38*
  48 HOURS  8.82
  51 HOURS  8.16

ANOTHER GAGE ?   Y OR N ?
```

Figure 4

```
STOP AT 3260
*RUN

 TYPE 4 PRECIP AMOUNTS (6 HRLY)  BEGINNING WITH THE EARLIEST ? .5,.75,0,0
IF YOU WANT ONLY HIGH STAGES AND CRESTS TYPE   HIGH.
   IF YOU WANT ALL STAGES TYPE ALL ? HIGH
WHICH BASIN DO YOU WANT FORECASTS FOR?
  1 = ALL
  2 = CHEMUNG  4 = BUFFALO AREA CREEKS
  3 = GENESEE  5 = ALLEGHENY
 ? 4
WHICH GAGE DO YOU WANT FORECASTS FOR?
   1 = ALL              23 = LANCASTER
  20 = BATAVIA          24 = GARDENVILLE
  21 = WILLIAMSVILLE    25 = EBENEZER
  22 = SCAJAQUADA       26 = GOWANDA
 ? 20
TYPE STAGE AT START OF PRECIP ? 3.55
TYPE RUNOFF INDEX AT LANCASTER ? 20
TONAWANDA CREEK AT BATAVIA   FLOOD STAGE 9 FEET

  BASE STAGE  3.55   FLOOD INDEX 18.18

  6 HRLY PRECIP  0.50  0.75  0.00  0.00

  TIMES OF STAGES ARE FROM THE START OF THE FIRST 6 HOUR PERIOD


  CREST  6.40 AT 33 HOURS


ANOTHER GAGE ?   Y OR N ?
```

Figure 5

```
STOP AT 3260
*RUN

 TYPE 4 PRECIP AMOUNTS (6 HRLY)  BEGINNING WITH THE EARLIEST ? .5,1,.5,0
IF YOU WANT ONLY HIGH STAGES AND CRESTS TYPE   HIGH.
   IF YOU WANT ALL STAGES TYPE ALL ? HIGH
WHICH BASIN DO YOU WANT FORECASTS FOR?
  1 = ALL
  2 = CHEMUNG  4 = BUFFALO AREA CREEKS
  3 = GENESEE  5 = ALLEGHENY
 ? 2
WHICH GAGE DO YOU WANT FORECASTS FOR?
  1 = ALL            5 = ERWINS
  2 = CAMPBELL       6 = CORNING
  3 = WEST CAMERON   7 = ELMIRA
  4 = LINDLEY        8 = CHEMUNG
 ? 1
TYPE THE FLOOD INDEX IF IT IS THE SAME FOR ALL GAGES.
  IF SOME ARE DIFFERENT TYPE 999 ? 999
TYPE STAGE AT START OF PRECIP,  FINAL INDEX
  CAMPBELL ? .74,3.15
  WEST CAMERON ? 3.82,2.9
  ERWINS ? 1.47,3.2

  6 HRLY PRECIP  0.50  1.00  0.50  0.00

  TIMES OF STAGES ARE FROM THE START OF THE FIRST 6 HOUR PERIOD

COHOCTON RIVER AT CAMPBELL    FLOOD STAGE 8 FEET
  BASE STAGE   .74   FLOOD INDEX  3.15

  CREST  5.15 AT 21 HOURS


CANISTEO RIVER AT WEST CAMERON   FLOOD STAGE 17 FEET
  BASE STAGE  3.82   FLOOD INDEX  2.90

  CREST 10.76 AT 18 HOURS


TIOGA RIVER AT LINDLEY IS CONTROLLED BY THE TIOGA DAM...NO FORECAST AVAILABLE


TIOGA RIVER AT ERWINS   FLOOD STAGE 18 FEET
  BASE STAGE  1.47   FLOOD INDEX  3.20

  CREST  8.04 AT 30 HOURS


CHEMUNG RIVER AT CORNING   FLOOD STAGE 29 FEET
   DIKES PROTECT CITY TO 39 FEET
   CALL CORNING GLASS AT 21 FEET  DAY 607-974-7238  NIGHT 607-974-7504

  CREST 20.31 AT 32 HOURS


CHEMUNG RIVER AT ELMIRA   FLOOD STAGE 10 FEET
   DIKES PROTECT CITY TO 23.6 FEET

  CREST  6.70 AT 36 HOURS
```

```
CHEMUNG RIVER AT CHEMUNG   FLOOD STAGE 16 FEET
  CREST 10.47 AT 41 HOURS

ANOTHER BASIN?  Y OR N ? N
NEW PRECIP AMOUNTS?  Y OR N ? Y
SAME GAGES?...Y OR N ? Y
TYPE 4 PRECIP AMOUNTS (6 HRLY)  BEGINNING WITH THE EARLIEST ? .75,1,.75,.25
IF YOU WANT ONLY HIGH STAGES AND CRESTS TYPE   HIGH.
   IF YOU WANT ALL STAGES TYPE ALL ? HIGH
COHOCTON RIVER AT CAMPBELL   FLOOD STAGE 8 FEET
  BASE STAGE   .74   FLOOD INDEX  3.15

  21 HOURS  6.75
  24 HOURS  6.98
  27 HOURS  6.71


CANISTEO RIVER AT WEST CAMERON   FLOOD STAGE 17 FEET
  BASE STAGE  3.82   FLOOD INDEX  2.90

  CREST 12.75 AT 24 HOURS


TIOGA RIVER AT LINDLEY IS CONTROLLED BY THE TIOGA DAM...NO FORECAST AVAILABLE


TIOGA RIVER AT ERWINS   FLOOD STAGE 18 FEET
  BASE STAGE  1.47   FLOOD INDEX  3.20

  CREST 11.15 AT 30 HOURS


CHEMUNG RIVER AT CORNING   FLOOD STAGE 29 FEET
   DIKES PROTECT CITY TO 39 FEET
    CALL CORNING GLASS AT 21 FEET  DAY 607-974-7238  NIGHT 607-974-7504

  26 HOURS 21.12
  29 HOURS 21.81
  32 HOURS 22.38
  35 HOURS 22.15
  38 HOURS 21.70
  41 HOURS 21.01


CHEMUNG RIVER AT ELMIRA   FLOOD STAGE 10 FEET
   DIKES PROTECT CITY TO 23.6 FEET

  33 HOURS  8.65
  36 HOURS  9.39
  39 HOURS  9.10
  42 HOURS  8.52


CHEMUNG RIVER AT CHEMUNG   FLOOD STAGE 16 FEET

  41 HOURS 13.56
  44 HOURS 13.23


ANOTHER BASIN?  Y OR N ? N
NEW PRECIP AMOUNTS?  Y OR N ? N

END AT 3350
```

Figure 5, Continued

C. Output Produced:

Output is produced on the Dasher and consists of the stages in feet for 3-hour periods if all stages were requested, otherwise only high stages are printed. If a stage does not meet a threshold, only the crest is printed. Stages over flood stage are flagged with an asterisk. See sample runs, Figures 2-5.

D. Cautions/Restrictions:

Only 24 hours of precipitation values can be entered because of the method used to compute runoff. Precipitation must be fairly uniform over the whole basin to accurately forecast downstream points. Since the three-hour hydrographs have been derived from and adjusted to six-hour hydrographs, the forecast stages for rainfall durations over six hours will tend to be too high.

This program is intended for small basins and headwaters areas. Accuracy will diminish with increasing basin size due to effects of computing mean areal precipitation, streamflow routing from upper portions of the basin and effects of upstream reservoirs.

The program keeps the base flow constant. If the base flow is too high, recession will not be accounted for and the later stages will be too high.

E. Future Development:

Addition of a recession factor would enable computation of stages over extended periods provided flood indices were recomputed.

Also, graphic presentation of the hydrographs is envisioned. Then the forecaster could plot observed versus forecast stages and adjust the forecast crest.

F. Adaptation to Other Stations:

Adapting HYDRO to run at another station will take quite a bit of work. The Data General manuals listed under References will be of some help, as will the list of variables included with the program listing. The following elements must be altered:

1. The high stage and flood stage arrays F4 and F2 must be changed. The high stage array is used to decide whether stages should be printed out under the high stage only option. The flood stage array is used to flag forecast flood stages with an asterisk. There is one high stage and one flood stage indicator per gage used, and the number associated with the gage indicates the position of its values in the respective arrays (e.g., Batavia is station 2Ø; the 20th element of F4 holds its high stage value and the 20th element of F2 holds its flood stage value).

2. The names of the basins and gages will have to be changed in the print statements in the input and output sections. Also, the flow of logic depends on the variables I2 (basin number) and I3 and I5 (gage numbers), so the program may have to be restructured to reflect changes in their possible values.

3. Each new basin must have its precip-index table inserted (RI table). Each new gage must have 48 points from its 3-hour hydrograph and 2Ø pairs of points from its rating table inserted in the program (see data section at end of program).

G. Complete Program Listing

VARIABLES:

B$	ANOTHER BASIN?
B1	BASE STAGE
B2	BASE FLOW
B3	BASE FLOW MATRIX
F$	ASTERISK
F	FLOOD INDEX
F1	BASIN FLOOD INDEX
F2	FLOOD STAGES
F3	FLOOD INDEX MATRIX
F4	HIGH STAGE MATRIX
F5	BASIN INDEX MATRIX
G$	ANOTHER GAGE?
G1	LAG MATRIX
H	UNIT HYDROGRAPH MATRIX
I	LOOP INDEX
I1	PRECIPITATION LOOP INDEX
I2	BASIN NUMBER
I3	GAGE NUMBER
I4	GAGE LOOP INDEX
I5	GAGE NUMBER
I6	ALL BASINS
J	LOOP INDEX
K	LOOP INDEX
L	LOWER RUNOFF TABLE VALUES
P1	INPUT PRECIPITATION
P2	3 HOURLY PRECIPITATION
P	PRECIPITATION SUMS
Q$	NEW PRECIPITATION?
Q2$	HIGH STAGES ONLY?
R1	RUNOFF VALUES
R2	COMPUTED PERIOD RUNOFF
R3	PRECIPITATION RATIO
R4	RUNOFF FOR LOWER BOUND
R5	LOWER RUNOFF RATIO
R6	RUNOFF FROM UPPER BOUND
R7	UPPER RUNOFF RATIO
S	STAGE INCREMENTS FROM RATING TABLE
S1	3 HOURLY STAGES
S2	CORNING STAGES
S3	CREST
T	PRECIPITATION INCREMENTS
U	UPPER RUNOFF TABLE VALUES
V	STAGE PRINTED
V1	TIME OF CREST
W	FLOW INCREMENTS OF RATING TABLE
W1	3 HOURLY FLOWS

```
0010 REM PROGRAM TO COMPUTE 3 HOUR STAGES IN WESTERN NEW YORK
0020 REM      WRITTEN BY ALAN BLACKBURN   FEBRUARY 5, 1981
0030 PRINT
0040 DIM R1[8],W[8,2],U[8,2]
0050 DIM S2[48],G1[48]
0060 DIM B3[35],F3[35],F4[35]
0070 DIM P[8],P1[4],P2[8]
0080 DIM S[20],W[20],F2[35]
0090 DIM H[48],W1[48],S1[48]
0100 FOR I=2 TO 32
0110   READ F4[I]
0120 NEXT I
0130 DATA 6.5,14,14,15,21,8,13,0,0,0,0,8,14,4,4
0140 DATA 28,11,0,7.5,6.5,8,6.5,5.5,9,8
0150 DATA 0,0,0,0,8,11.5
0160 PAGE =130
0170 PRINT "TYPE 4 PRECIP AMOUNTS (6 HRLY)  BEGINNING WITH THE EARLIEST";
0180 INPUT P1[1],P1[2],P1[3],P1[4]
0190 PRINT "IF YOU WANT ONLY HIGH STAGES AND CRESTS TYPE   HIGH."
0200 PRINT "   IF YOU WANT ALL STAGES TYPE ALL";
0210 INPUT Q2$
0220 IF Q$="Y" THEN GOTO 0610
0230 LET I6=0
0240 RESTORE 0290
0250 REM INPUT FLOOD STAGES (F2)
0260 FOR I=2 TO 32
0270   READ F2[I]
0280 NEXT I
0290 DATA 8,17,17,18,29,10,16,0,0,0,0,10,17,5.5
0300 DATA 33,13,0,9,8,10,8,7,11,10
0310 DATA 0,0,0,0,10,14
0320 PRINT "WHICH BASIN DO YOU WANT FORECASTS FOR?"
0330 PRINT "  1 = ALL"
0340 PRINT "  2 = CHEMUNG  4 = BUFFALO AREA CREEKS"
0350 PRINT "  3 = GENESEE  5 = ALLEGHENY"
0360 INPUT I2
0370 IF I2=1 THEN GOTO 0400
0380 PRINT "WHICH GAGE DO YOU WANT FORECASTS FOR?"
0390 ON I2 THEN GOTO 0400, 0770, 0640, 0720, 0550
0400 LET I6=1
0410 LET I3=1
0420 FOR I2=2 TO 5
0430   ON (I2-1) THEN GOSUB 0450, 0470, 0490, 0510
0440   ON (I2-1) THEN GOTO 0830, 0700, 0700, 0590
0450   PRINT "CHEMUNG BASIN"
0460   RETURN
0470   PRINT "GENESEE BASIN"
0480   RETURN
0490   PRINT "BUFFALO AREA CREEKS"
0500   RETURN
0510   PRINT "ALLEGHENY BASIN"
0520   RETURN
0530 NEXT I2
0540 GOTO 2870
0550 PRINT "   1 = ALL    32 = SALAMANCA"
```

```
0560 PRINT "  31 = OLEAN"
0570 INPUT I3
0580 IF I3<>1 THEN GOTO 3400
0590 PRINT "TYPE ANTECEDENT INDEX IF IT IS THE SAME FOR ALL GAGES"
0600 GOTO 0840
0610 IF I3<>1 THEN GOTO 3210
0620 IF I6=1 THEN GOTO 2880
0630 ON (I2-1) THEN GOTO 2970, 3180, 3100, 3140
0640 PRINT "   1 = ALL              16 = CHURCHVILLE"
0650 PRINT "  13 = WELLSVILLE       17 = AVON"
0660 PRINT "  14 = PORTAGEVILLE     18 = ROCHESTER"
0670 PRINT "  15 = GARBUTT"
0680 INPUT I3
0690 IF I3<>1 THEN GOTO 3390
0700 PRINT "TYPE RUNOFF INDEX IF IT IS THE SAME FOR ALL GAGES."
0710 GOTO 0840
0720 PRINT "   1 = ALL              23 = LANCASTER"
0730 PRINT "  20 = BATAVIA          24 = GARDENVILLE"
0740 PRINT "  21 = WILLIAMSVILLE    25 = EBENEZER"
0750 PRINT "  22 = SCAJAQUADA       26 = GOWANDA"
0760 GOTO 0680
0770 PRINT "  1 = ALL             5 = ERWINS"
0780 PRINT "  2 = CAMPBELL        6 = CORNING"
0790 PRINT "  3 = WEST CAMERON    7 = ELMIRA"
0800 PRINT "  4 = LINDLEY         8 = CHEMUNG"
0810 INPUT I3
0820 IF I3<>1 THEN GOTO 0890
0830 PRINT "TYPE THE FLOOD INDEX IF IT IS THE SAME FOR ALL GAGES."
0840 PRINT "  IF SOME ARE DIFFERENT TYPE 999";
0850 INPUT F1
0860 IF I6=1 THEN LET F5[I2]=F1
0870 GOTO 1900
0880 PRINT
0890 ON I3 THEN GOTO 1900, 3500, 3500, 3210, 3500, 3710, 3710, 3710
0900 REM RUNOFF SUBROUTINE
0910 REM CONVERT FROM 6 TO 3 HOUR PRECIP
0920 FOR I=1 TO 4
0930   LET P2[2*I]=P1[I]/2
0940   LET P2[2*I-1]=P1[I]/2
0950 NEXT I
0960 FOR I=1 TO 8
0970   LET P[I]=0
0980   REM SUM TOTAL PRECIP
0990   FOR I1=1 TO I
1000     LET P[I]=P[I]+P2[I1]
1010   NEXT I1
1020   ON I2 THEN GOSUB 1110, 1160, 1130, 1130, 1130
1030   IF I1=99 THEN GOTO 1100
1040   LET R1[I]=R2
1050   IF I=1 THEN GOTO 1100
1060   REM DETERMINE RUNOFF IN PERIOD BY SUBTRACTING PREVIOUS PERIODS
1070   FOR I1=2 TO I
1080     LET R1[I]=R1[I]-R1[I1-1]
1090   NEXT I1
1100 NEXT I
1110 RETURN
1120 REM GENESEE, BUF AREA, ALLEGHENY PRECIP COLUMNS
1130 RESTORE 1150
1140 GOTO 1180
1150 DATA .25,.5,1,1.5,2,2.5,3,4,5,6
```

```
1160 RESTORE 1220
1170 REM READ PRECIP COLUMNS IN PRECIP - RUNOFF, FI TABLE
1180 FOR J=1 TO 10
1190   READ T[J]
1200 NEXT J
1210 REM CHEMUNG PRECIP COLUMNS
1220 DATA .25,.5,1,1.5,2,3,4,4.75,5,6
1230 IF P[I]<.25 THEN GOTO 1290
1240 IF P[I]>=6 THEN GOTO 1370
1250 REM DETERMINE WHICH PRECIP COLUMNS TO INTERPOLATE BETWEEN
1260 FOR J=1 TO 9
1270   IF P[I]<T[J+1] THEN GOTO 1390
1280 NEXT J
1290 LET I1=0
1300 FOR J=1 TO 4
1310   LET I1=I1+P1[J]
1320 NEXT J
1330 IF I1>=.25 THEN LET I1=99
1340 IF I1=99 THEN GOTO 1360
1350 PRINT "PRECIP MUST BE MORE THAN .24 INCHES"
1360 RETURN
1370 PRINT "PRECIP MUST BE LESS THAN 6 INCHES"
1380 RETURN
1390 ON I2 THEN GOTO 1400, 1400, 1420, 1440, 1460
1400 RESTORE 1620
1410 GOTO 1490
1420 RESTORE 6610
1430 GOTO 1490
1440 RESTORE 7180
1450 GOTO 1490
1460 RESTORE 7730
1470 GOTO 1490
1480 REM READ IN PRECIP COLUMNS FOR INTERPOLATION
1490 FOR K=1 TO J
1500   FOR L1=1 TO 8
1510     FOR M=1 TO 2
1520       READ L[L1,M]
1530     NEXT M
1540   NEXT L1
1550 NEXT K
1560 FOR K=1 TO 8
1570   FOR L1=1 TO 2
1580     READ U[K,L1]
1590   NEXT L1
1600 NEXT K
1610 REM CHEMUNG FLOOD INDEX TABLE
1620 DATA 5,0,3,0,2.1,.02,1.6,.05,.6,.16,0,0,0,0,0,0
1630 DATA 5,0,3.6,.01,2.7,.03,1.8,.11,1.3,.2,.6,.35,0,0,0,0
1640 DATA 5,0,3.5,.04,2.8,.11,2.1,.24,1.6,.39,1.2,.53,.8,.69,.6,.76
1650 DATA 5,.01,3.7,.08,2.4,.35,1.6,.68,.9,1.04,.6,1.19,0,0,0,0
1660 DATA 5,.04,4.1,.11,3.3,.26,2.7,.45,2.1,.72,1.5,1.07,.6,1.64,0,0
1670 DATA 5,.16,4.2,.32,3.6,.48,3.2,.69,2.8,.9,2.3,1.21,1.9,1.5
1680 DATA .6,2.55
1690 DATA 5,.42,4.3,.67,3.6,1.01,2.7,1.59,2.4,1.82,1.7,2.44,.6,3.48,0,0
1700 DATA 5,.73,4.2,1.11,3.1,1.81,2.3,2.47,1.6,3.16,.6,4.18,0,0,0,0
1710 DATA 5,.88,3.9,1.44,3.8,1.44,3.2,2.11,2.6,2.55,2,3.07,.6,4.43,0,0
1720 DATA 5,1.34,3.9,2.14,3.8,2.52,3.1,3.01,2.1,3.88,.6,5.46,0,0,0,0
1730 LET R3=(P[I]-T[J])/(T[J+1]-T[J])
1740 REM ROMPUTE RUNOFF FOR LOWER BOUND R4
1750 FOR K=2 TO 8
```

```
1760   IF F>LC(K,1) THEN GOTO 1780
1770 NEXT K
1780 LET K=K-1
1790 LET R5=(F-LC(K+1),1))/(LC(K,1)-LC(K+1),1))
1800 LET R4=LC(K+1),2)-(LC(K+1),2)-LC(K,2))*R5
1810 REM COMPUTE RUNOFF FOR UPPER BOUND R6
1820 FOR K=2 TO 8
1830   IF F>UC(K,1) THEN GOTO 1850
1840 NEXT K
1850 LET K=K-1
1860 LET R7=(F-UC(K+1),1))/(UC(K,1)-UC(K+1),1))
1870 LET R6=UC(K+1),2)-(UC(K+1),2)-UC(K,2))*R7
1880 LET R2=R4+((R6-R4)*R3)
1890 RETURN
1900 IF F1=999 THEN GOTO 2390
1910 PRINT "TYPE STAGE AT START OF PRECIP"
1920 ON (I2-2) THEN GOTO 2020, 2160, 2330
1930 PRINT "  CAMPBELL";
1940 INPUT B3(2)
1950 PRINT "  WEST CAMERON";
1960 INPUT B3(3)
1970 PRINT "  ERWINS";
1980 INPUT B3(5)
1990 LET F=F1
2000 IF I6=1 THEN GOTO 0530
2010 GOTO 2950
2020 PRINT "  WELLSVILLE";
2030 INPUT B3(13)
2040 PRINT "  PORTAGEVILLE";
2050 INPUT B3(14)
2060 PRINT "  GARBUTT";
2070 INPUT B3(15)
2080 PRINT "  CHURCHVILLE";
2090 INPUT B3(16)
2100 PRINT "  AVON";
2110 INPUT B3(17)
2120 PRINT "  ROCHESTER";
2130 INPUT B3(18)
2140 IF I6=1 THEN GOTO 0530
2150 GOTO 3160
2160 PRINT "  BATAVIA";
2170 INPUT B3(20)
2180 PRINT "  WILLIAMSVILLE";
2190 INPUT B3(21)
2200 PRINT "  SCAJAQUADA";
2210 INPUT B3(22)
2220 PRINT "  LANCASTER";
2230 INPUT B3(23)
2240 IF F1=999 THEN GOTO 2490
2250 PRINT "  GARDENVILLE";
2260 INPUT B3(24)
2270 PRINT "  EBENEZER";
2280 INPUT B3(25)
2290 PRINT "  GOWANDA";
2300 INPUT B3(26)
2310 IF I6=1 THEN GOTO 0530
2320 GOTO 3080
2330 PRINT "  OLEAN";
2340 INPUT B3(31)
2350 PRINT "  SALAMANCA";
```

```
2360 INPUT B3[32]
2370 IF I6=1 THEN GOTO 0530
2380 GOTO 3120
2390 ON (I2-1) THEN GOTO 2400, 2490, 2650, 2790
2400 PRINT "TYPE STAGE AT START OF PRECIP,  FINAL INDEX"
2410 PRINT "  CAMPBELL";
2420 INPUT B3[2],F3[2]
2430 PRINT "  WEST CAMERON";
2440 INPUT B3[3],F3[3]
2450 PRINT "  ERWINS";
2460 INPUT B3[5],F3[5]
2470 IF I6=1 THEN GOTO 0530
2480 GOTO 2950
2490 PRINT "TYPE STAGE AT START OF PRECIP,  RUNOFF INDEX"
2500 IF I2=4 THEN GOTO 2710
2510 PRINT "  WELLSVILLE";
2520 INPUT B3[13],F3[13]
2530 PRINT "  PORTAGEVILLE";
2540 INPUT B3[14],F3[14]
2550 PRINT "  GARBUTT";
2560 INPUT B3[15],F3[15]
2570 PRINT "  CHURCHVILLE";
2580 INPUT B3[16],F3[16]
2590 PRINT "  AVON";
2600 INPUT B3[17],F3[17]
2610 PRINT "  ROCHESTER";
2620 INPUT B3[18],F3[18]
2630 IF I6=1 THEN GOTO 0530
2640 GOTO 3160
2650 PRINT "TYPE RUNOFF INDEX FOR LANCASTER";
2660 INPUT F3[23]
2670 LET F3[20]=F3[23]
2680 LET F3[21]=F3[23]
2690 LET F3[22]=F3[23]
2700 GOTO 1910
2710 PRINT "  GARDENVILLE";
2720 INPUT B3[24],F3[24]
2730 PRINT "  EBENEZER";
2740 INPUT B3[25],F3[25]
2750 PRINT "  GOWANDA";
2760 INPUT B3[26],F3[26]
2770 IF I6=1 THEN GOTO 0530
2780 GOTO 3080
2790 PRINT "TYPE STAGE AT START OF PRECIP,  ANTECEDENT INDEX"
2800 PRINT "  OLEAN";
2810 INPUT B3[31],F3[31]
2820 PRINT "  SALAMANCA";
2830 INPUT B3[32],F3[32]
2840 IF I6=1 THEN GOTO 0530
2850 GOTO 3120
2860 REM ITERATION FOR COMPUTING ALL GAGES AND PRINTING
2870 PRINT
2880 FOR I2=2 TO 5
2890   LET F1=F5[I2]
2900   PRINT
2910   ON (I2-1) THEN GOSUB 0450, 0470, 0490, 0510
2920   ON (I2-1) THEN GOTO 2950, 3160, 3080, 3120
2930 NEXT I2
2940 GOTO 3320
2950 PRINT
```

```
2960 GOSUB 5250
2970 FOR I4=2 TO 8
2980   LET B1=B3[I4]
2990   LET F=F1
3000   IF F1=999 THEN LET F=F3[I4]
3010   ON I4 THEN GOSUB 3350, 3550, 3630, 3590, 3570, 3650, 3750, 3790
3020   ON (I4-12) THEN GOSUB 3820, 3840, 3860, 3880, 3900, 3920
3030   ON (I4-19) THEN GOSUB 3940, 3990, 4040, 4090, 4110, 4130, 4150
3040   ON (I4-30) THEN GOSUB 4170, 4200
3050 NEXT I4
3060 IF I6=1 THEN GOTO 2930
3070 GOTO 3290
3080 PRINT
3090 GOSUB 5250
3100 FOR I4=20 TO 26
3110   GOTO 2980
3120   PRINT
3130   GOSUB 5250
3140   FOR I4=31 TO 32
3150     GOTO 2980
3160     PRINT
3170     GOSUB 5250
3180     FOR I4=13 TO 18
3190       GOTO 2980
3200       REM SINGLE GAGE SELECTION
3210       ON I3 THEN GOSUB 3350, 3550, 3630, 3590, 3570, 3650, 3750, 3790
3220       ON (I3-12) THEN GOSUB 3820, 3840, 3860, 3880, 3900, 3920
3230       ON (I3-19) THEN GOSUB 3940, 3990, 4040, 4090, 4110, 4130, 4150
3240       ON (I3-30) THEN GOSUB 4170, 4200
3250       PRINT "ANOTHER GAGE ?   Y OR N";
3260       INPUT G$
3270       IF G$="N" THEN GOTO 3290
3280       GOTO 0380
3290       PRINT "ANOTHER BASIN?  Y OR N";
3300       INPUT B$
3310       IF B$="Y" THEN GOTO 0320
3320       PRINT "NEW PRECIP AMOUNTS?  Y OR N";
3330       INPUT Q$
3340       IF Q$="Y" THEN GOTO 3360
3350       END
3360       PRINT "SAME GAGES?...Y OR N";
3370       INPUT Q$
3380       GOTO 0170
3390       ON (I2-1) THEN GOTO 3510, 3420, 3440, 3400
3400       PRINT "TYPE STAGE AT START OF PRECIP,  ANTECEDENT INDEX"
3410       GOTO 3520
3420       PRINT "TYPE STAGE AT START OF PRECIP,  RUNOFF INDEX"
3430       GOTO 3520
3440       ON (I3-19) THEN GOTO 3450, 3450, 3450, 3420, 3420, 3420, 3420
3450       PRINT "TYPE STAGE AT START OF PRECIP";
3460       INPUT B1
3470       PRINT "TYPE RUNOFF INDEX AT LANCASTER";
3480       INPUT F
3490       GOTO 3230
3500       IF Q$="Y" THEN GOTO 3210
3510       PRINT "TYPE STAGE AT START OF PRECIP,  FINAL INDEX"
3520       INPUT B1,F
3530       GOSUB 5210
3540       GOTO 3210
3550       PRINT "COHOCTON RIVER AT CAMPBELL    FLOOD STAGE 8 FEET"
```

```
3560        GOTO 4230
3570        PRINT "TIOGA RIVER AT ERWINS   FLOOD STAGE 18 FEET"
3580        GOTO 4230
3590        PRINT "TIOGA RIVER AT LINDLEY IS CONTROLLED BY THE TIOGA DAM!!!NO FORECAST AVAILABLE"
3600        PRINT
3610        PRINT
3620        RETURN
3630        PRINT "CANISTEO RIVER AT WEST CAMERON   FLOOD STAGE 17 FEET"
3640        GOTO 4230
3650        PRINT "CHEMUNG RIVER AT CORNING   FLOOD STAGE 29 FEET"
3660        PRINT "   DIKES PROTECT CITY TO 39 FEET"
3670        PRINT "   CALL CORNING GLASS AT 21 FEET  DAY 607-974-7238  NIGHT 607-974-7504"
3680        IF I3=1 THEN GOTO 5300
3690        PRINT
3700        GOTO 4230
3710        PRINT "TYPE FLOOD INDEX FOR ERWINS AND ERWINS STAGE AT START OF PRECIP";
3720        INPUT F,B1
3730        GOSUB 5210
3740        GOTO 3210
3750        PRINT "CHEMUNG RIVER AT ELMIRA   FLOOD STAGE 10 FEET"
3760        PRINT "   DIKES PROTECT CITY TO 23.6 FEET"
3770        IF I3=1 THEN GOTO 5470
3780        GOTO 4230
3790        PRINT "CHEMUNG RIVER AT CHEMUNG   FLOOD STAGE 16 FEET"
3800        IF I3=1 THEN GOTO 5630
3810        GOTO 4230
3820        PRINT "GENESEE RIVER AT WELLSVILLE   FLOOD STAGE 10 FEET"
3830        GOTO 4230
3840        PRINT "GENESEE RIVER AT PORTAGEVILLE   FLOOD STAGE 17 FEET"
3850        GOTO 3810
3860        PRINT "OATKA CREEK AT GARBUTT   FLOOD STAGE 5 FEET"
3870        GOTO 4230
3880        PRINT "BLACK CREEK AT CHURCHVILLE   FLOOD STAGE 5 FEET"
3890        GOTO 4230
3900        PRINT "GENESEE RIVER AT AVON   FLOOD STAGE 33 FEET"
3910        GOTO 4230
3920        PRINT "GENESEE RIVER AT ROCHESTER   FLOOD STAGE 13 FEET"
3930        GOTO 4230
3940        PRINT "TONAWANDA CREEK AT BATAVIA   FLOOD STAGE 9 FEET"
3950        LET F=F/1.1
3960        IF F<=10 THEN LET F=10.001
3970        IF I3<>1 THEN GOSUB 5210
3980        GOTO 4230
3990        PRINT "ELLICOTT CREEK AT WILLIAMSVILLE   FLOOD STAGE 8 FEET"
4000        LET F=F*1.28
4010        IF F>=80 THEN LET F=79.999
4020        IF I3<>1 THEN GOSUB 5210
4030        GOTO 4230
4040        PRINT "SCAJAQUADA CREEK   FLOOD STAGE 10 FEET"
4050        LET F=F/1.1*.57
4060        IF F<=10 THEN LET F=10.001
4070        IF I3<>1 THEN GOSUB 5210
4080        GOTO 4230
4090        PRINT "CAYUGA CREEK AT LANCASTER   FLOOD STAGE 8 FEET"
4100        GOTO 4230
4110        PRINT "BUFFALO CREEK AT GARDENVILLE   FLOOD STAGE 7 FEET"
4120        GOTO 4230
4130        PRINT "CAZENOVIA CREEK AT EBENEZER   FLOOD STAGE 11 FEET"
4140        GOTO 4230
4150        PRINT "CATTARAUGUS CREEEK AT GOWANDA   FLOOD STAGE 10 FEET"
```

```
4160        GOTO 4230
4170        PRINT "ALLEGHENY RIVER AT OLEAN   FLOOD STAGE 10 FEET"
4180        PRINT "   DIKES PROTECT CITY TO 24.5 FEET"
4190        GOTO 4230
4200        PRINT "ALLEGHENY RIVER AT SALAMANCA   FLOOD STAGE 14 FEET"
4210        PRINT "   DIKES PROTECT CITY TO 22 FEET"
4220        GOTO 4230
4230        REM COMPUTE RUNOFF
4240        IF I3<>1 THEN GOTO 4260
4250        PRINT USING "  BASE STAGE---.##   FLOOD INDEX ##.##",B1,F
4260        GOSUB 0920
4270        REM READ 3 HRLY UNIT HYDROGRAPH (H), STAGE (S), FLOW (W) INCREMENTS
4200        ON I3 THEN GOTO 4320, 4360, 4420, 4400, 4380, 4380, 4380, 4380
4290        ON (I3-12) THEN GOTO 4460, 4480, 4500, 4520, 4540, 4560
4300        ON (I3-19) THEN GOTO 4580, 4600, 4620, 4640, 4660, 4680, 4700
4310        ON (I3-30) THEN GOTO 4720, 4740
4320        ON I4 THEN GOTO 3350, 4360, 4420, 4400, 4380, 5300, 5470, 5630
4330        ON (I4-12) THEN GOTO 4460, 4480, 4500, 4520, 4540, 4560
4340        ON (I4-19) THEN GOTO 4580, 4600, 4620, 4640, 4660, 4680, 4700
4350        ON (I4-30) THEN GOTO 4720, 4740
4360        RESTORE 6380
4370        GOTO 4760
4380        RESTORE 6460
4390        GOTO 4760
4400        RESTORE 6530
4410        GOTO 4760
4420        RESTORE 6540
4430        GOTO 4760
4440        RESTORE 6600
4450        GOTO 4760
4460        RESTORE 6720
4470        GOTO 4760
4480        RESTORE 6790
4490        GOTO 4760
4500        RESTORE 6870
4510        GOTO 4760
4520        RESTORE 6950
4530        GOTO 4760
4540        RESTORE 7040
4550        GOTO 4760
4560        RESTORE 7110
4570        GOTO 4760
4580        RESTORE 7290
4590        GOTO 4760
4600        RESTORE 7360
4610        GOTO 4760
4620        RESTORE 7430
4630        GOTO 4760
4640        RESTORE 7490
4650        GOTO 4760
4660        RESTORE 7550
4670        GOTO 4760
4680        RESTORE 7610
4690        GOTO 4760
4700        RESTORE 7670
4710        GOTO 4760
4720        RESTORE 7840
4730        GOTO 4760
4740        RESTORE 7920
4750        GOTO 4760
```

```
4760      FOR J=1 TO 48
4770        READ H[J]
4780      NEXT J
4790      FOR J=1 TO 20
4800        READ S[J]
4810        READ W[J]
4820      NEXT J
4830      REM COMPUTE BASE FLOW (B2)
4840      IF B1>S[1] THEN GOTO 4870
4850      LET B2=W[1]
4860      GOTO 4930
4870      FOR J=2 TO 20
4880        IF B1<S[J] THEN GOTO 4920
4890      NEXT J
4900      PRINT "BASE STAGE TOO HIGH"
4910      RETURN
4920      LET B2=((B1-S[J-1])/(S[J]-S[J-1]))*(W[J]-W[J-1])+W[J-1]
4930      REM COMPUTE 3 HOURLY FLOWS
4940      LET W1[1]=B2+R1[1]*H[1]
4950      LET W1[2]=B2+R1[1]*H[2]+R1[2]*H[1]
4960      LET W1[3]=B2+R1[1]*H[3]+R1[2]*H[2]+R1[3]*H[1]
4970      LET W1[4]=B2+R1[1]*H[4]+R1[2]*H[3]+R1[3]*H[2]+R1[4]*H[1]
4980      LET W1[5]=B2+R1[1]*H[5]+R1[2]*H[4]+R1[3]*H[3]+R1[4]*H[2]+R1[5]*H[1]
4990      LET W1[6]=B2+R1[1]*H[6]+R1[2]*H[5]+R1[3]*H[4]+R1[4]*H[3]+R1[5]*H[2]+R1[6]*H[1]
5000      LET W1[7]=B2+R1[1]*H[7]+R1[2]*H[6]+R1[3]*H[5]+R1[4]*H[4]+R1[5]*H[3]+R1[6]*H[2]+R1[7]*H[1]
5010      FOR J=8 TO 48
5020        LET W1[J]=B2+R1[1]*H[J]+R1[2]*H[J-1]+R1[3]*H[J-2]+R1[4]*H[J-3]
5030        LET W1[J]=W1[J]+R1[5]*H[J-4]+R1[6]*H[J-5]+R1[7]*H[J-6]+R1[8]*H[J-7]
5040      NEXT J
5050      REM COMPUTE STAGES (S1)
5060      FOR J=1 TO 48
5070        IF W1[J]>W[1] THEN GOTO 5100
5080        LET S1[J]=S[1]
5090        GOTO 5180
5100        FOR K=2 TO 20
5110          IF W1[J]<W[K] THEN GOTO 5160
5120          IF K=20 THEN GOTO 5140
5130          IF W[K+1]=0 THEN GOTO 5160
5140        NEXT K
5150        PRINT "FLOW EXCEEDS RATING TABLE AND FLOOD OF RECORD "
5160        LET S1[J]=((W1[J]-W[K-1])/(W[K]-W[K-1]))*(S[K]-S[K-1])+S[K-1]
5170        LET G1[J]=0
5180      NEXT J
5190      ON (I3-5) THEN GOTO 5300, 5300, 5300
5200      GOTO 5810
5210      REM HEADING SUBROUTINE
5220      PRINT
5230      PRINT USING "  BASE STAGE---.##   FLOOD INDEX ##.##",B1,F
5240      PRINT
5250      PRINT USING "  6 HRLY PRECIP  #.##  #.##  #.##  #.##",P1[1],P1[2],P1[3],P1[4]
5260      PRINT
5270      PRINT "  TIMES OF STAGES ARE FROM THE START OF THE FIRST 6 HOUR PERIOD"
5280      PRINT
5290      RETURN
5300      REM COMPUTE CORNING - ERWINS STAGE RELATIONSHIP
5310      FOR J=1 TO 48
5320        LET G1[J]=2
5330        IF S1[J]>=20 THEN GOTO 5360
5340        LET S1[J]=(S1[J]-2)/18*11.95+16.3
5350        GOTO 5430
```

```
5360        IF S1[J]>=22.3 THEN GOTO 5390
5370        LET S1[J]=(S1[J]-20)/2.3*2.25+28.25
5380        GOTO 5430
5390        IF S1[J]>=24.6 THEN GOTO 5420
5400        LET S1[J]=(S1[J]-22.3)/2.3*4.6+30.5
5410        GOTO 5430
5420        LET S1[J]=(S1[J]-24.6)/2*5.9+36.1
5430      NEXT J
5440      IF I3=7 THEN GOTO 5470
5450      IF I3=8 THEN GOTO 5630
5460      GOTO 5810
5470      REM COMPUTE CORNING - ELMIRA STAGE RELATIONSHIP
5480      FOR J=1 TO 48
5490        LET S2[J]=S1[J]
5500        IF S1[J]>=24 THEN GOTO 5530
5510        LET S1[J]=(S1[J]-18)/6*7.8+3.7
5520        GOTO 5600
5530        IF S1[J]>=28.8 THEN GOTO 5560
5540        LET S1[J]=(S1[J]-24)/4.8*5.9+11.5
5550        GOTO 5600
5560        IF S1[J]>=36 THEN GOTO 5590
5570        LET S1[J]=(S1[J]-28.8)/7.2*5+17.4
5580        GOTO 5600
5590        LET S1[J]=(S1[J]-36)/4.8*3.1+22.4
5600        LET G1[J]=6
5610      NEXT J
5620      GOTO 5810
5630      REM COMPUTE CORNING - CHEMUNG STAGE RELATIONSHIP
5640      FOR J=1 TO 48
5650        IF I3=8,THEN LET S2[J]=S1[J]
5660        IF S2[J]>=20 THEN GOTO 5690
5670        LET S1[J]=(S2[J]-18)/2*2+8
5680        GOTO 5760
5690        IF S2[J]>=24 THEN GOTO 5720
5700        LET S1[J]=(S2[J]-20)/4*6+10
5710        GOTO 5760
5720        IF S2[J]>=32.5 THEN GOTO 5750
5730        LET S1[J]=(S2[J]-24)/8.5*8.1+16
5740        GOTO 5760
5750        LET S1[J]=(S2[J]-32.5)/8.2*7.4+24.1
5760        LET G1[J]=11
5770        IF S1[J]>16 THEN LET G1[J]=13
5780        IF S1[J]>20 THEN LET G1[J]=15
5790        IF S1[J]>23 THEN LET G1[J]=16
5800      NEXT J
5810      PRINT
5820      REM PRINT STAGES
5830      LET I5=I3
5840      IF I3=1 THEN LET I5=I4
5850      IF Q2$="HIGH" THEN GOTO 6230
5860      FOR J=1 TO 8
5870        LET F$=" "
5880        IF S1[J]>=F2[I5] THEN LET F$="*"
5890        PRINT USING "  ## HOURS---.##",(3*J+G1[J]),S1[J];
5900        PRINT F$;
5910        LET F$=" "
5920        IF S1[J+8]>=F2[I5] THEN LET F$="*"
5930        PRINT USING "  ## HOURS---.##",(3*(J+8)+G1[J+8]),S1[J+8];
5940        PRINT F$;
5950        LET F$=" "
```

```
5960        IF S1[J+16]>=F2[I5] THEN LET F$="*"
5970        PRINT USING " ### HOURS---.##",(3*(J+16)+G1[J+16]),S1[J+16];
5980        PRINT F$;
5990        IF I5<4 THEN GOTO 6210
6000        ON (I5-19) THEN GOTO 6010, 6010, 6210, 6210, 6210, 6210, 6210
6010        LET F$=" "
6020        IF S1[J+24]>=F2[I5] THEN LET F$="*"
6030        PRINT USING " ### HOURS---.##",(3*(J+24)+G1[J+24]),S1[J+24];
6040        PRINT F$;
6050        ON (I5-14) THEN GOTO 6080, 6210, 6210, 6080, 6210, 6080
6060        ON (I5-30) THEN GOTO 6080, 6080
6070        GOTO 6210
6080        LET F$=" "
6090        IF S1[J+32]>=F2[I5] THEN LET F$="*"
6100        PRINT USING " ### HOURS---.##",(3*(J+32)+G1[J+32]),S1[J+32];
6110        PRINT F$;
6120        GOTO 6210
6130        LET F$=" "
6140        IF S1[J+40]>=F2[I5] THEN LET F$="*"
6150        PRINT USING " ### HOURS ##.##",(3*(J+40)+G1[J+40]),S1[J+40];
6160        PRINT F$
6170      NEXT J
6180      PRINT
6190      PRINT
6200      RETURN
6210      PRINT
6220      GOTO 6170
6230      LET S3=0
6240      LET V=0
6250      FOR J=1 TO 48
6260        LET F$=" "
6270        IF S1[J]>F2[I5] THEN LET F$="*"
6280        IF S1[J]>S3 THEN LET S3=S1[J]
6290        IF S1[J]=S3 THEN LET V1=J*3+G1[J]
6300        IF S1[J]<F4[I5] THEN GOTO 6340
6310        LET V=1
6320        PRINT USING " ### HOURS ##.##",((3*J)+G1[J]),S1[J];
6330        PRINT F$
6340      NEXT J
6350      IF V=1 THEN GOTO 6180
6360      PRINT USING "  CREST **.** AT### HOURS";S3;V1
6370      GOTO 6180
6380      REM CAMPBELL DATA
6390      REM CAMPBELL DATA
6400      DATA 6.2,10.9,16.6,14.6,11.8,8.9,7.7,5.3,4.3,3.2
6410      DATA 2.2,1.5,1.1,.8,.6,.5,.5,.4,0,0
6420      DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
6430      DATA .1,.353,1,2.13,2.1,.86,2.6,1.29,3.1,1.82,5,4.1,5.4,4.66
6440      DATA 5.8,5.3,6.3,6.2,6.8,7.24,7.2,8.24,7.5,9.1,7.9,10.42,8.3,11.9
6450      DATA 8.8,13.9,9.9,19.6,10.7,25.7,11,29.3,11.2,32.7,11.6,41.1
6460      REM ERWINS DATA
6470      DATA 3,6.5,13,20,42.5,40.7,37,33.5,18.2,14
6480      DATA 11.7,9.3,7.5,5.8,5.2,4.5,4.2,3.5,3.2,2.5
6490      DATA 2,2,1.5,1.5,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
6500      DATA .8,.05,2.4,.91,3.8,2.48,5.5,5.25,8.4,12,11.4,21,12.6,25.2
6510      DATA 13.6,29.2,15.2,36.4,16.6,43.4,17,45.6,17.4,48,17.8,50.6
6520      DATA 19,59,20,66.5,23.9,90,24.6,108,25.2,124,26.1,155,26.8,193.5
6530      REM LINDLEY DATA
6540      REM WEST CAMERON DATA
6550      DATA 5,8.5,18.3,16.9,5.5,3.8,3.3,2.3,2.1,2
```

```
6560        DATA 2,1.5,.5,0,0,0,0,0,0,0
6570        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
6580        DATA 3,.015,5.7,.975,7.3,2.03,9,3.4,10.5,5,12.5,7.45,13.5,8.9,14.8,11.04,16.2,13.63,17.6,16.5
6590        DATA 18.1,17.6,18.5,18.66,19,20.2,20,23.7,21,27.75,21.5,30,22,32.5,22.5,35.5,23,39,23.5,43
6600        REM CORNING DATA
6610        REM GENESEE RI TABLE
6620        DATA 80,0,28.5,0,15,.05,10,.1,0,0,0,0,0,0,0,0
6630        DATA 80,0,35,0,23,.05,17,.1,10,.23,0,0,0,0,0,0
6640        DATA 80,0,62.7,0,43,.05,35.25,.1,29.8,.15,26.75,.2,16.3,.4,10,.6
6650        DATA 80,.01,53.6,.05,38.9,.15,27.67,.4,21.33,.6,16.33,.8,13,.95,10,1.1
6660        DATA 80,.03,50.33,.15,37,.29,32,.45,25.88,.7,20.63,.95,13.14,1.4,10,1.6
6670        DATA 80,.06,50,.23,45,.27,40,.36,32,.67,25,1.06,15,1.73,10,2.1
6680        DATA 80,.09,50,.34,45,.41,40,.52,35,.72,25,1,38,13,2,35,10,2,6
6690        DATA 80,.23,60,.53,55,.61,45,.83,40,.98,35,1.23,30,1.62,10,3.6
6700        DATA 80,.41,50,1.245,45,1.41,40,1.62,38,1.73,35,1.97,30,2.44,10,4.6
6710        DATA 80,.83,55,1.77,50,1.98,46,2.17,43,2.31,39,2.57,35,2.91,10,5.6
6720        REM WELLSVILLE DATA
6730        DATA 1.6,2.6,11.6,11.1,6.4,4.2,3.1,2.5,2,1.8,1.4,1.4,1.2,1.2
6740        DATA 1,1,1,.8,.8,.8,.8,.4,.4,.4,.4,.4,.4,.4,.4,.4,.4,.4
6750        DATA .2,.2,.2,.2,.2,.2,0,0,0,0,0,0,0,0
6760        DATA 4,.016,5,.3,6.1,1.19,6.9,2.12,7.5,2.9,8.1,3.74,8.5,4.34,8.8,4.82
6770        DATA 9.2,5.5,9.5,6.04,9.9,6.8,10.3,7.6,10.7,8.44,11.1,9.32,11.5,10.24
6780        DATA 11.9,11.2,12.3,12.2,12.7,13.24,13.4,15.2,20.7,20.1
6790        REM PORTAGEVILLE DATA
6800        DATA .9,1.1,13,15.8,21,23,23.6,22.6,12.5,10,7.6
6810        DATA 6.3,5.3,4.7,4.3,3.8,3.4,3.1,2.8,2.5
6820        DATA 2.1,1.9,1.8,1.7,1.6,1.5,1.4,1.3
6830        DATA 1.2,1.1,1,1,1,.9,.8,.8,.7,.7,.6,.6
6840        DATA 0,0,0,0,0,0,0,0
6850        DATA 8,.06,9.2,.46,10,1.14,10.9,2.35,11.6,4.01,26,47.7,28,54.6,32,70.59
6860        DATA 34.4,82.49,38,103.37,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
6870        REM GARBUTT DATA
6880        DATA .1,.1,.5,.8,1.45,1.6,1.75,1.95,2.15,2.35
6890        DATA 3,2.7,2.35,2.15,1.95,1.75,1.35,1.15,.95,.85
6900        DATA .8,.8,.7,.65,.55,.5,.45,.35,.35,.3
6910        DATA .25,.2,.2,.2,.2,.2,.2,.2,.2,.2
6920        DATA .1,.1,.1,.1,.1,.1,.1,.1
6930        DATA 2,0,3,.2,3,5,.4,4,5,1,5,1,5,6,2,7,7,4,8,5,6,9,9,9,7
6940        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
6950        REM CHURCHVILLE DATA
6960        DATA .1,.1,.55,.7,.85,1,1.1,1.15,1.2,1.25
6970        DATA 1.75,1.65,1.65,1.5,1.45,1.4,1.25,1.1,.8,.7
6980        DATA .7,.65,.65,.55,.45,.35,.3,.3,.2,.2
6990        DATA .2,.2,.2,.2,.2,.2,.2,.2,.1,.1
7000        DATA .1,.1,.1,.1,.1,.1,.1,.1
7010        DATA 1,0,1.5,.02,2,.08,2.7,.21,3,.28,3.6,.46,4.2,.69,5,1.04
7020        DATA 6,1.6,7,2.3,8,3.25,9,4.35,9.5,4.95,10.9,7.05
7030        DATA 0,0,0,0,0,0,0,0,0,0,0,0
7040        REM AVON DATA
7050        DATA 1.5,2.6,3.5,4.6,6.15,5.6,3.25,3.1,3,2.6
7060        DATA 2.2,1.9,1.7,1.5,1.3,1.1,1,.8,.75,.6,.55,.4,.3,.1
7070        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7080        DATA 14,.1,15.4,.5,17.8,1.38,19.6,2.09,23.6,4,26,5.4,29,7.5,32,9.85
7090        DATA 35,12.9,38.2,17,42.1,23,49,29.99
7100        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7110        REM ROCHESTER DATA
7120        DATA .8,1.7,2.5,3.1,3.9,4.5,5.3,5.9,6.4,6.7
7130        DATA 6.8,6.85,6.8,6.8,6.7,6.4,6.1,5.8,5.7,5.4
7140        DATA 5.1,4.6,4.4,4.4,4.15,4.1,3.8,3.7,3.4,3.1,2.7
7150        DATA 2.65,2.4,2.2,2.1,1.9,1.8,1.7,1.6,1.4,1.3,1.1,0,0,0,0,0,0,0,0
```

```
7160        DATA .1,0,1,.02,3,1.09,5,2.67,7,5.11,8,6.79,10,11,12,16.3,15,26.1,23.9,59.4
7170        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7180        REM BUFFALO AREA RI TABLE
7190        DATA 80,0,30,0,15,.05,10,.1,0,0,0,0,0,0,0,0
7200        DATA 80,0,40,0,15,.13,10,.23,0,0,0,0,0,0,0,0
7210        DATA 80,0,60,0,30,.17,18,.33,15,.44,10,.61,0,0,0,0
7220        DATA 80,.01,40,.14,30,.3,25,.43,20,.57,17,.67,13,.86,10,1.03
7230        DATA 80,.03,40,.22,30,.43,25,.6,20,.8,17,.96,13,1.21,10,1.45
7240        DATA 80,.07,40,.38,30,.62,25,.81,20,1.08,17,1.29,13,1.65,10,1.95
7250        DATA 80,.11,40,.54,30,.81,25,1.03,20,1.36,17,1.64,13,2.09,10,2.45
7260        DATA 80,.23,45,.84,35,1.14,30,1.375,25,1.69,20,2.18,15,2.77,10,3.45
7270        DATA 80,.41,45,1.41,35,1.84,30,2.13,25,2.55,20,3.13,15,3.75,10,4.45
7280        DATA 80,.83,45,2.2,35,2.72,30,3.05,27,3.32,24,3.63,20,4.1,10,5.45
7290        REM BATAVIA DATA
7300        DATA 0,0,.2,.5,1.2,2.1,2.6,2.9,3.2,3
7310        DATA 2.6,2.3,2,1.7,1.4,1.1,.85,.7,.65,.5
7320        DATA .45,.4,.4,.4,.3,.3,.2,.2,.1,.1
7330        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7340        DATA 1.7,0,5.5,1.91,10,4.54,11,5.35,12,6.4,13,7.56,13.9,8.9
7350        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7360        REM WILLIAMSVILLE DATA
7370        DATA .4,.55,.55,.7,1,1.3,1.6,1.75,1.75,1.8,1.4,.9,.65,.45,.35,.2,.1,.1
7380        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7390        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7400        DATA 1.5,.038,2.2,.135,2.7,.235,3.2,.36,4.6,.78,7.4,1.76,9,2.4
7410        DATA 10,5.4,0,0,0,0
7420        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7430        REM SCAJAQUADA DATA
7440        DATA .54,.96,.73,.49,.27,.19,.09,.06..03,.01,0,0,0,0,0,0,0,0
7450        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7460        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7470        DATA 1.6,0,1.95,.032,2.5,.125,3.1,.245,5.1,.6,10.5,1.23,14.38,2.62,0,0,0,0,0,0
7480        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7490        REM REM LANCASTER DATA
7500        DATA .15,.6,7.6,6,1.7,.9,.45,.35,.3,.27,.24,.19,.14,.05
7510        DATA 0,0,0,0,0,0,0,0,0,0
7520        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7530        DATA 3,0,4,.04,5,.46,5.6,.98,6.6,2.3,8.5,6,10,8.6,12,11.9,13,13.6,18,22.1
7540        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7550        REM GARDENVILLE DATA
7560        DATA .15,.5,10.5,8.8,2.9,1.8,1,.6,.5,.4,.3,.2,.1,.05
7570        DATA 0,0,0,0,0,0,0,0,0,0
7580        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7590        DATA 1,0,2,.28,3,.78,4,1.6,5,3.1,6,5.02,7,6.95,9,11.8,10,14.47,0,0
7600        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7610        REM EBENEZER DATA
7620        DATA .2,1.1,8.8,8.2,3.2,2,1.2,.9,.85,.7,.65,.55,.5,.45
7630        DATA .35,.3,.3,.3,.25,.2,.2,.2,.2,.2,.2,.2,.2
7640        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7650        DATA 2,.01,2.5,.09,3,.21,5.1,1.42,7,2.95,9,4.9,11,7.4,13,10.4
7660        DATA 16.15.6,17,17.9,17.1,18.23,18,21.13,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7670        REM GOWANDA DATA
7680        DATA .6,1.5,16,13.9,8.6,7.1,3.3,2.7,2.4,2.1,1.8,1.5,1.4,1.2,1.15
7690        DATA 1.05,1,1,.9,.9,.8,.8,.7,.7,.6,.6,.5,.5,.4,.4
7700        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7710        DATA 2,.2,4,1.6,6,4.8,7,7.2,9,12.8,11,20.3,13,29,15,39.2,18,56.37,0,0
7720        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7730        REM ALLEGHENY RI TABLE
7740        DATA 7.5,0,2,0,1.5,.07,1.2,.15,0,0,0,0,0,0,0,0
7750        DATA 7.5,0,3.5,0,2.5,.06,2,.16,1.2,.4,0,0,0,0,0,0
```

```
7760        DATA 7.5,0,2.9,.17,2.6,.25,2.3,.33,2,.4,1.5,.63,1.2,.8,0,0
7770        DATA 7.5,0,5,.03,4,.11,3.5,.19,2,.7,1.7,.88,1.4,1.14,1.2,1.31
7780        DATA 7.5,0.5,.07,4,.23,3.5,.38,3,.57,2.3,.08,2,1.15,1.2,1.81
7790        DATA 7.5,0,6,.01,4.5,.23,3.6,.55,3,.8,2.5,1.17,2,1.6,1.2,2.3
7800        DATA 7.5,0,5.5,.1,4.9,.23,4,.56,3.5,.8,3,1.17,2.5,1.6,1.2,2.79
7810        DATA 7.5,.03,5.5,.25,5,.46,4.5,.72,4,1.08,3.5,1.59,3,2.02,1.2,3.77
7820        DATA 7.5,.06,6,.3,5.6,.47,5.3,.71,5,.92,4.5,1.38,3.5,2.39,1.2,4.74
7830        DATA 7.5,.09,7,.16,6.5,.31,6.2,.44,5.9,.65,5.7,.85,5.5,1.09,1.2,5.69
7840        REM OLEAN DATA
7850        DATA .7,1.3,2.3,3.4,8,10,12.5,12.2,11.9,11.3
7860        DATA 9.5,8.5,6.2,5.3,4.7,4.1,2.5,1.9,1.7,1.45
7870        DATA 1.4,1.15,1.05,.9,.8,.5,.4,.1,.1,.1
7880        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7890        DATA -1.9,.11,-.8,.21,3,2.9,6,6.05,7.5,7.8,8,8.7
7900        DATA 10,12.2,14.4,21,20,40,25,66,30,100,32.9,124
7910        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
7920        REM SALAMANCA DATA
7930        DATA 1.2,1.8,7.3,7.9,8.8,10.2,13.8,15.3,17.7,18.3
7940        DATA 18.7,18.2,17.6,16.8,15.7,14.7,14.2,13.3,11.4,10.8
7950        DATA 9.6,8.7,7.9,7.3,6.6,6,5.5,4.9,4.5,3.9
7960        DATA 3.4,3.1,2.8,2.4,2.2,1.95,1.75,1.6,1.55,1.4
7970        DATA 1,.8,.55,.45,.4,.3,0,0
7980        DATA 2.5,0,3,.16,3.5,.56,4,1.12,4.6,2.04,5.1,2.94,5.5,4.14,6,4.8
7990        DATA 8,9.8,9,14.1,13.8,29.2,17,42.8,24.7
8000        DATA 0,0,0,0,0,0,0,0,0,0,0,0,0
```